
Overview
    

LiveView Native is a framework for building native applications using Elixir and Phoenix LiveView.
It upgrades your existing Phoenix project with the ability to write truly native user interfaces that
run on devices and platforms beyond the web.
Here's an example of a simple LiveView that renders both HTML and native SwiftUI views:

  
    
  
  hello_live.ex


# lib/my_app_web/live/hello_live.ex
defmodule MyAppWeb.HelloLive do
  use Phoenix.LiveView
  use MyAppWeb, :live_view
  use MyAppWeb.HelloStyles

  @impl true
  def render(%{format: :swiftui} = assigns) do
    # This render function serves native SwiftUI views
    # It uses the `~SWIFTUI` sigil instead of `~H`
    ~SWIFTUI"""
    <VStack spacing={8}>
      <Button phx-click="hello">
        <Image class="fg-color-purple font-size-48 p-8" system-name="sparkles"></Image>
      </Button>
      <HStack spacing={4}>
        Hello world on <Text class="bold"><%= device_name(assigns) %></Text>!
      </HStack>
    </VStack>
    """
  end

  @impl true
  def render(%{} = assigns) do
    # This render function serves HTML
    ~H"""
    <div class="flex w-full h-screen items-center">
      <ul class="w-full text-center">
        <li>
          <button phx-click="hello">
            <.icon name="hero-sparkles-solid" class="text-purple-500 w-24 h-24 m-4" />
          </button>
        </li>
        <li>Hello world on <span class="font-bold"><%= device_name(assigns) %></span>!</li>
      </ul>
    </div>
    """
  end

  @impl true
  def handle_event("hello", _params, socket) do
    # This event handler can be shared across all platforms
    IO.puts "Hello world!"

    {:noreply, socket}
  end

  ###

  # This function can be called from both HTML and SwiftUI templates
  # The native device type is available as the `@target` assign
  defp device_name(%{target: :phone}), do: "iOS"
  defp device_name(%{target: :pad}), do: "iPadOS"
  defp device_name(%{target: :mac}), do: "macOS"
  defp device_name(%{target: :watch}), do: "watchOS"
  defp device_name(_), do: "the web"
end

  
    
  
  hello_styles.ex


# lib/my_app_web/live/hello_styles.ex
defmodule MyAppWeb.HelloStyles do
  use LiveViewNative.Stylesheet, :swiftui

  ~SHEET"""
  "bold" do
    fontWeight(.bold)
  end

  "fg-color-" <> color do
    foregroundStyle(to_ime(color))
  end

  "font-size-" <> font_size do
    font(system(size: to_integer(font_size)))
  end

  "p-" <> padding do
    padding(to_integer(padding))
  end
  """

  def class(_other, _), do: {:unmatched, ""}
end
This code serves users whether they're using a web browser or native app running on an iPhone,
iPad, macOS desktop or Apple Watch. Each platform renders its own native widgets and UI elements,
allowing state, event callbacks and business logic to be shared.

  
    
  
  iPhone


[image: Hello World - iPhone]

  
    
  
  iPad


[image: Hello World - iPad]

  
    
  
   Watch


[image: Hello World - macOS]

  
    
  
  Desktop (macOS)


[image: Hello World - macOS]

  
    
  
  Desktop (Web)


[image: Hello World - Web]
The following native platforms are officially supported, and support for other platforms
can be provided by third-party platform libraries.
	iOS 16+
	macOS 13+
	watchOS 9+
	Android

LiveView Native requires some foundational knowledge to use. You should already be familiar with
Elixir, the Phoenix Framework and
Phoenix LiveView. If you're looking to learn
more about any of these subjects, there are a lot of great resources available. Some recommended
materials include the Elixir guides,
Elixir learning resources page, Phoenix guides,
Phoenix community page and the Phoenix LiveView HexDocs. 
With those prerequisites out of the way, let's get LiveView Native installed!



  

    
Installation
    

There are a number of steps that must be completed before you can use LiveView Native. This document covers them, which are as follows:
	Setup your Phoenix project with the minimum required versions of Elixir, Phoenix and LiveView
	Add LiveView Native to your mix.exs dependencies
	Add any number of platform libraries to enable support for native clients
	Configure your project to use LiveView Native
	Update your application's web module, router and layouts


  
    
  
  1. Prepare your Phoenix app


To use LiveView Native, you must have an existing Phoenix project. If you don't have one, follow the Up and Running section of the official Phoenix guide to create one.
Next, make sure your project meets the following requirements:
	it uses Elixir 1.15 or greater — to enforce this, update your mix.exs:
def project do
[
  # ...
  elixir: "~> 1.15",
  # ...
]
end

	it uses Phoenix >= 1.7 and Phoenix LiveView >= 0.18 — your mix.exs should partially resemble this:
def deps do
[
  {:phoenix, "~> 1.7"},
  {:phoenix_live_view, "~> 0.18"},
  # other dependencies here...
]
end


If your project doesn't meet this criteria then it will need to be upgraded. Some helpful resources for this include the Elixir installation page, Phoenix installation page, as well as the changelogs for Elixir, Phoenix and Phoenix LiveView.

  
    
  
  2. Add LiveView Native


Once you've met the requirements to use LiveView Native, simply add it to your list of dependencies in your project's mix.exs:
def deps do
  [
    # other dependencies here...
    {:live_view_native, "~> 0.2"}
  ]
end

  
    
  
  3. Add platform libraries


The :live_view_native dependency isn't useful on its own. You'll also need to add any platform libraries you want your project to be compatible with. These libraries provide the platform-specific code that allows them to connect to your app and render LiveViews within their native environments.
This guide covers installation for the officially supported platforms, SwiftUI (iOS, macOS and watchOS) and Jetpack (Android).

  
    
  
  SwiftUI


Adds compatibility for iOS 16+, macOS 13+ and watchOS 9+.
def deps do
  [
    # other dependencies here...
    {:live_view_native_swiftui, "~> 0.2"}
  ]
end

  
    
  
  Jetpack


Adds compatibility for Android.
Warning
The Jetpack client is under development and is not yet usable in end-user applications.
For more information, check the repo for live_view_native_jetpack here.

def deps do
  [
    # other dependencies here...
    {:live_view_native_jetpack, "~> 0.0.0"}
  ]
end
Make sure to fetch any dependencies you've added to your mix.exs before continuing:
mix deps.get


  
    
  
  4. Configure LiveView Native


LiveView Native includes a Mix task that can automatically handle the process of configuring your project to support it. If that is not to your liking, manual setup is also an option. This guide includes instructions for both of these approaches.

  
    
  
  Guided (recommended)


Within your project directory, run the following command:
mix lvn.install

This command will prompt you to answer a few questions. If everything goes well, you should see a message that your project has been configured to use LiveView Native.

  
    
  
  Manual


After adding the :live_view_native Hex package and any platform libraries, define a key-value configuration in your Config file (this is typically found in config/config.exs). This configuration expects a :plugins option which takes a list of any platform libraries you want your application to support. Platform libraries are represented by their top-level namespace module:
# config.exs

# Use LiveView Native to add support for native platforms
config :live_view_native,
  plugins: [
    # other plugins here...
    LiveViewNative.SwiftUI,
    LiveViewNative.Jetpack
  ]

# LiveView Native Stylesheet support
# Omit this if you're not using platforms that support LiveView
# Native stylesheets
config :live_view_native_stylesheet,
  parsers: [
    swiftui: LiveViewNative.SwiftUI.RulesParser
  ]
Next, create a project for each platform's native client using the official tools provided by each platform.
This step varies across platforms — for example, SwiftUI development uses Xcode
and so you'll need to know how to create a new Xcode project,
add liveview-client-swiftui as a package dependency, configure your device targets
and write the glue code in Swift to connect to your Phoenix application. For Jetpack this process involves working with
Android Studio and Kotlin to create the client, since those are the tools relevant
to Android development.
Manual configuration at this point is outside of the scope of these HexDocs but each platform library should have more
information within its own documentation. To find that, consult the official repositories for each platform library
that you need to create a native client app for.

  
    
  
  5. Update your application code


After running mix lvn.install to configure the LiveView Native framework, you will need to integrate it with
your application. This will involve making the following changes to your Phoenix application's code:
	Using LiveViewNative.LiveView and LiveViewNative.LiveComponent in your web module
	Adding the LiveViewNative.SessionPlug plug to your router's :browser pipeline
	Allowing Phoenix to render native layouts by using LiveViewNative.Layouts in your layouts module and scoping embed_templates/1 to *.html instead of *

These changes are required to use LiveView Native. The following subsections describe these changes in more detail
and how to properly apply them.

  
    
  
  5a. Use LiveView Native in your web module


Most Phoenix-related modules in your application will use a web module that was automatically
generated for you when running mix phx.gen. By default, this module might be named something like
MyAppWeb and would be found at lib/my_app_web/my_app_web.ex (replacing my_app with the actual
name of your app). This is your app's web module.
Within the web module, you should find functions called live_view and live_component. Update these
functions so that inside each function's quote block it contains use LiveViewNative.LiveView and
use LiveViewNative.LiveComponent respectively. Be sure to add only those two lines without changing
or deleting anything else:
defmodule MyAppWeb do
  @moduledoc """
  The entrypoint for defining your web interface, such
  as controllers, components, channels, and so on...
  """

  # (truncated for example purposes...)

  def live_view do
    quote do
      use Phoenix.LiveView,
        layout: {MyAppWeb.Layouts, :app}

      unquote(html_helpers())

      # LiveView Native support
      use LiveViewNative.LiveView
    end
  end

  def live_component do
    quote do
      use Phoenix.LiveComponent

      unquote(html_helpers())

      # LiveView Native support
      use LiveViewNative.LiveComponent
    end
  end

  # (truncated for example purposes...)
end
These two lines upgrade LiveViews and LiveComponents in your app with the ability to render native
UI code in addition to HTML. Once these changes are saved, move on to the second part of this section
which is updating your app's router to be able to handle native sessions.

  
    
  
  5b. Add the LiveView Native session plug to your router.ex


Your Phoenix router defines the various paths to resources within your app, such as
API endpoints, static pages and of course LiveViews. Native clients that connect to
your app also pass through the router and you need to ensure that it can properly
handle these non-web connections.
Update your router (located somewhere like lib/my_app_web/router.ex) so that the 
:browser pipeline near the top of the file contains the LiveViewNative.SessionPlug:
defmodule MyAppWeb.Router do
  use MyAppWeb, :router

  pipeline :browser do
    plug :accepts, ["html"]
    plug :fetch_session
    plug :fetch_live_flash

    # (truncated for example purposes...)

    # LiveView Native support
    plug LiveViewNative.SessionPlug
  end

  # (truncated for example purposes...)
end
With this plug, your router is now able to understand the initial request payloads that LiveView
Native clients send when connecting to your app. With that change to your router saved, you can now
move on to the third and final step of this installation guide, updating your layout module. 

  
    
  
  5c. Allow your app to render native layouts


Phoenix applications are made up of two layouts, a root layout and an app layout. Typically these
layouts are HTML-based which only works in a web environment. To avoid layout templates for the web
from rendering in non-web environments and to allow writing native layouts, you'll need to update
Phoenix's root layout module so that it's compatible with LiveView Native.
Open your app's main layout module, typically located at lib/my_app_web/components/layouts.ex,
and change any embed_templates lines so that the catch-all wildcard * is replaced with one that
only checks for files with the *.html extension. Also, add use LiveViewNative.Layouts near the
top of the file to add support for native layouts:
defmodule MyAppWeb.Layouts do
  use MyAppWeb, :html
  use LiveViewNative.Layouts

  # changed from `embed_templates "layouts/*"`
  embed_templates "layouts/*.html"
end
If you have templates for other file extensions, like .json or .txt, you can add separate embed_templates
lines for those as well. The important point is to avoid the catch-all * as it will cause the default Phoenix
template system to try to compile HEEx templates with non-HTML syntax like files ending in .swiftui.heex and
.jetpack.heex. This will cause an error like this which prevents your app from compiling: 
== Compilation error in file lib/my_app_web/components/layouts.ex ==
** (Phoenix.LiveView.Tokenizer.ParseError) lib/my_app_web/components/layouts/root.swiftui.heex:1:1: invalid tag <NavigationStack>
  |
1 | <NavigationStack>
  | ^
    (phoenix_live_view 0.19.5) lib/phoenix_live_view/tag_engine.ex:1391: Phoenix.LiveView.TagEngine.raise_syntax_error!/3
    (phoenix_live_view 0.19.5) lib/phoenix_live_view/tag_engine.ex:452: Phoenix.LiveView.TagEngine.handle_token/2
    (elixir 1.15.6) lib/enum.ex:2510: Enum."-reduce/3-lists^foldl/2-0-"/3
    (phoenix_live_view 0.19.5) lib/phoenix_live_view/tag_engine.ex:182: Phoenix.LiveView.TagEngine.handle_body/1
    (phoenix_live_view 0.19.5) expanding macro: Phoenix.LiveView.HTMLEngine.compile/1
    lib/my_app_web/components/layouts/root.swiftui.heex: LvnExampleWeb.Layouts.root/1
Once your layout module has been updated appropriately, save the changes and continue.

  
    
  
  Post-Installation


If you've reached this section, you should have already done the following:
	Added :live_view_native and any platform libraries to your list of dependencies in mix.exs.
	Run mix lvn.install or configured LiveView Native manually.
	Updated your web module, router and layouts module with LiveView Native specific code.

If all of these steps have been completed properly, you should be able to run your Phoenix app in development as usual:
iex -S mix phx.server

To confirm that LiveView Native has been properly installed, call the following function in an IEx
session. It should return a map describing all of the native platform libraries that have been installed
for your app.
iex(1)> LiveViewNative.platforms()
Confirm that each platform listed (excluding html) has a platform-specific client app (i.e. Xcode,
Android Studio, etc.) for connecting to your LiveView Native backend. If you used mix lvn.install
to enable LiveView Native, these project files will be placed in the native/ directory of your app.  
If everything looks good, continue on to writing your first native LiveView.



  

    
Your First Native LiveView
    

Now that you've installed LiveView Native, its time to create a LiveView
that uses it. This guide explains how to create a basic LiveView module that can serve different
templates to both web and native platforms while sharing non-template code between them.

  
    
  
  A basic LiveView


Info
For more information on LiveView in general, see the docs for Phoenix.LiveView here.

To begin, create a new LiveView module and add it to your Phoenix router. The path for this LiveView
should be identical to the path you chose when setting up your native client app (this defaults to "/",
when using mix lvn.install to install LiveView Native).
First create a new LiveView somewhere in your app that inherits LiveViewNative.LiveView, like so:
# lib/my_app_web/live/hello_live.ex
defmodule MyAppWeb.HelloLive do
  use Phoenix.LiveView
  use MyAppWeb, :live_view

  @impl true
  def render(%{} = assigns) do
    # This UI renders on the web
    ~H"""
    <div class="flex w-full h-screen items-center">
      <span class="w-full text-center">
        Hello web!
      </span>
    </div>
    """
  end
end
Then, add it to your app's router:
# lib/my_app_web/router.ex
defmodule MyAppWeb.Router do
  use MyAppWeb, :router

  ### Pipelines, etc.
  
  scope "/", MyAppWeb do
    pipe_through :browser

    ### Other routes...

    live "/", HelloLive
  end

  ### Routes for LiveDashboard, Swoosh, etc.
end
Then, run your app and navigate to your local development server in a web browser.
You should see the LiveView you created.
[image: Hello World - Web]

  
    
  
  A native LiveView


To support non-web platforms using LiveView Native, simply provide a template for each
platform you want to support. This is done by extending LiveView's standard render/1 callback.
Each platform will have its own custom :format, render sigil and HEEx syntax, which are
all pulled in by the LiveViewNative.LiveView macro inherited toward the top of your LiveView.
The following example demonstrates handling the :swiftui platform for LiveView Native apps
with the :live_view_native_swiftui dependency, which covers iOS, iPadOS and macOS support:
# lib/my_app_web/live/hello_live.ex
defmodule MyAppWeb.HelloLive do
  use Phoenix.LiveView
  use MyAppWeb, :live_view

  @impl true
  def render(%{format: :swiftui} = assigns) do
    # This UI renders on the iPhone / iPad app
    ~SWIFTUI"""
    <VStack>
      <Text>
        Hello native!
      </Text>
    </VStack>
    """
  end

  @impl true
  def render(%{} = assigns) do
    # This UI renders on the web
    ~H"""
    <div class="flex w-full h-screen items-center">
      <span class="w-full text-center">
        Hello web!
      </span>
    </div>
    """
  end
end
Loading the native project at native/swiftui/MyApp/MyApp.xcodeproj and running it in
Simulator will render the native template as a native SwiftUI app:
[image: Hello World - iOS]
If everything looks as it should, congratulations! Your app is now using LiveView Native to
serve multiple platform-specific templates to both web and non-web clients.

  
    
  
  What's next?


If you've reached this section of the guide, you should have a cursory understanding of LiveView
Native, how to install it in an existing Phoenix project and how to write a basic "Hello World"
LiveView that renders a different UI for web and non-web clients.
To use LiveView Native to its full potential, you will need to dig deeper into the following subjects:
	LiveView Native features and concepts
	Additional libraries for the platform(s) your app supports
	Building UIs for the platform(s) your app support

This guide will cover the first point in the pages to follow. For the other two, you will need to consult the
relevant documentation, guides and supporting literature for the platform(s) you want your app to serve. Below
are some recommended resources for the SwiftUI and Jetpack platforms:

  
    
  
  SwiftUI


Covers iOS 16+, macOS 13+ and watchOS 9+.
	Platform library HexDocs
	Swift library docs
	SwiftUI docs
	SwiftUI tutorials
	Xcode docs


  
    
  
  Jetpack


Covers Android.
	Platform library HexDocs
	Jetpack Compose tutorial




  

    
Troubleshooting
    

Sometimes you might get blocked by a compiler error or other issue during installation. If this happens, you might try clearing your build directory and recompiling from a clean state. To do this, run the following commands from your project's root directory:
rm -rf _build
rm -rf deps
mix deps.get
mix compile

This should ideally resolve without error. If it doesn't, feel free to submit an issue on the GitHub repo for this library or ask in the #liveview-native channel of Elixir Slack; a member of the LiveView Native core team or other community member should be able to help you troubleshoot the problem.



  

    
Template Syntax
    

LiveView Native platforms use EEx templates which resemble the native APIs they're based
on as closely as possible. This is in contrast to frameworks that provide custom element
names which then get translated to native components.
Consider the following example for a SwiftUI view. It is represented in LiveView Native using
all of the same elements like List, HStack, etc. To do this we adapt the original SwiftUI
element and attribute names to the semantics of the EEx syntax using ~SWIFTUI, like so:

  
    
  
  Elixir


LiveView Native components for SwiftUI use SwiftUI-like syntax.
defmodule MyAppWeb.MyComponents do
  use Phoenix.LiveView
  use LiveViewNative.Component

  def album_detail(%{format: :swiftui} = assigns) do
    ~SWIFTUI"""
    <List>
      <%= for song <- @album.songs do %>
        <HStack>
          <Image name={@album.cover} />
          <VStack alignment="leading">
            <Text class="bold"><%= song.title %></Text>
            <Text><%= song.artist.name %></Text>
          </VStack>
        </HStack>
      <% end %>
    </List>
    """
  end
end

  
    
  
  Swift


The same component would be represented in SwiftUI like this:
import SwiftUI

struct AlbumDetail: View {
  var album: Album

  var body: some View {
    List(album.songs) { song in
      HStack {
        Image(album.cover)
        VStack(alignment: .leading) {
          Text(song.title)
            .fontWeight(.bold)
          Text(song.artist.name)
        }
      }
    }
  }
}
Here we can observe various semantic changes to port SwiftUI code to EEx:
	SwiftUI view names are used as elements, like List and VStack.	The argument syntax (i.e. alignment: .leading to alignment="leading") is adapted as well.


	The struct value album is an assign, @album, instead.
	Instead of passing album.songs as an argument, we use a comprehension.	<Text> elements take their arguments as values, similar to HTML.



These conventions can generally be applied to all sorts of examples when using LiveView Native to build SwiftUI views.
Because LiveView Native is modular, each platform library will have its own way of "bridging the gap" between the Elixir
side of your app and the native side.
Officially supported platform libraries are designed with general platform-parity in mind; while some light abstractions
may be used in cases where the native syntax doesn't map perfectly onto LiveView templates, any names of elements,
attributes, structs, enums, types, etc. are typically carried over in identical or approximate (i.e. camel case to
snake case) form.
For more information on writing platform-specific template code, consult the documentation for each platform you want
your app to support.

  
    
  
  SwiftUI


Covers iOS 16+, macOS 13+ and watchOS 9+.
	Platform library HexDocs
	Swift library docs
	SwiftUI docs
	SwiftUI tutorials
	Xcode docs


  
    
  
  Jetpack


Covers Android.
	Platform library HexDocs
	Jetpack Compose tutorial




  

    
Render Patterns
    

There are multiple ways to render native UIs with LiveView Native. This document covers various
patterns and when you might use them.

  
    
  
  render/1 function clauses


The most common pattern that is often used throughout this guide is the function clause pattern.
Each function clause matches on the :format, allowing each platform to define its own UI:
# lib/my_app_web/live/hello_live.ex
defmodule MyAppWeb.HelloLive do
  use Phoenix.LiveView
  use MyAppWeb, :live_view

  @impl true
  def render(%{format: :swiftui} = assigns) do
    # This UI renders on the iPhone / iPad app
    ~SWIFTUI"""
    <VStack>
      <Text>
        Hello native!
      </Text>
    </VStack>
    """
  end

  @impl true
  def render(%{} = assigns) do
    # This UI renders on the web
    ~H"""
    <div class="flex w-full h-screen items-center">
      <span class="w-full text-center">
        Hello web!
      </span>
    </div>
    """
  end
end

  
    
  
  Function components


Platform-specific components can also be defined. For example:
# lib/my_app_web/live/hello_live.ex
defmodule MyAppWeb.SharedComponents do
  use Phoenix.Component
  use LiveViewNative.Component

  def logo(%{format: :swiftui} = assigns) do
    ~SWIFTUI"""
    <VStack>
      <Image name="Logo" />
      <Text><%= @title %></Text>
    </VStack>
    """
  end

  def logo(%{} = assigns) do
    ~H"""
    <div>
      <img src="my-logo.png" />
      <h1><%= @title %></h1>
    </div>
    """
  end
end

  
    
  
  External template files


If you would prefer to break your render function out into separate template files, you can
conditionally render platform-specific templates using the render_native/1 macro. External
template files are namespaced according to their :format:

  
    
  
  hello_live.ex


defmodule MyAppWeb.HelloLive do
  use Phoenix.LiveView
  use MyAppWeb, :live_view

  @impl true
  def render(%{} = assigns) do
    render_native(assigns)
  end
end

  
    
  
  hello_live.html.heex


<div id="template-web">
  <div class="text-slate-800 bg-slate-50 h-screen w-screen grid grid-cols-1 gap-1 content-center items-center text-center">
    <div class="font-semibold mb-1">A web template, courtesy of hello_live.html.heex</div>
  </div>
</div>

  
    
  
  hello_live.swiftui.heex


<VStack id="template-ios">
  <HStack class="p-5">
    <Text>A SwiftUI template, courtesy of hello_live.swiftui.heex</Text>
  </HStack>
</VStack>

  
    
  
  Targeting specific devices


Conditional rendering based on device type is also supported, depending on the platform library you're using.
The following example for the SwiftUI platform renders different text on various devices:
# lib/my_app_web/live/hello_live.ex
defmodule MyAppWeb.SharedComponents do
  use Phoenix.Component
  use LiveViewNative.Component

  def logo(%{format: :swiftui} = assigns) do
    ~SWIFTUI"""
    <VStack>
      <%= case @target do %>
        <% :mac -> %>
          <Text>Hello macOS!</Text>
        <% :pad -> %>
          <Text>Hello iPadOS!</Text>
        <% :watch -> %>
          <Text>Hello watchOS!</Text>
        <% :tv -> %>
          <Text>Hello tvOS!</Text>
        <% _ -> %>
          <Text>Hello iOS!</Text>
      <% end %>
    </VStack>
    """
  end
end



  

    
Handling Events
    

Events are a common part of any LiveView application. On the server, LiveView Native handles events identically to LiveView
on the web. How the client handles events depends on the native platform and its respective client implementation. This
document provides some general information on events regardless of platform.

  
    
  
  Translated Bindings


Platforms may translate Phoenix bindings to their native counterparts
where it makes sense. The most common example of this is phx-click which listens for tap events on SwiftUI and Jetpack
targets.
On the server, events are handled using standard LiveView callbacks like handle_event/3 regardless of what platform they
come from. Callbacks can also be shared across platforms, assuming the events and their params are compatible. Here is a
basic example of sharing a handle_event/3 callback in this way:
# hello_live.ex
defmodule MyAppWeb.HelloLive do
  use Phoenix.LiveView
  use MyAppWeb, :live_view

  @impl true
  def mount(_params, _session, socket) do
    {:ok, assign(socket, name: "World")}
  end

  @impl true
  def render(%{format: :swiftui} = assigns) do
    # This UI renders on the iPhone / iPad app
    ~SWIFTUI"""
    <VStack>
      <Text phx-click="click_test">Hello <%= @name %>!</Text>
    </VStack>
    """
  end

  @impl true
  def render(%{} = assigns) do
    # This UI renders on the web
    ~H"""
    <div class="flex w-full h-screen items-center">
      <span class="w-full text-center" phx-click="click_test">
        Hello <%= @name %>!
      </span>
    </div>
    """
  end

  @impl true
  def handle_event("click_test", _params, socket) do
    # This event can be called from both SwiftUI and the web
    {:noreply, assign(socket, name: "José")}
  end
end
In this example, clicking the "Hello World" button changes it to say "Hello José" on both the web and iOS.
The same handle_event/3 callback is fired in both cases.

  
    
  
  Change Events


In LiveView Native, elements can support client-side changes to their value outside of a <form>.
Synchronizing the values must be handled manually by the LiveView using change events.

  
    
  
  Client-side changes


Use the phx-change attribute to respond to client-side changes to an element's value.
<TextField text={@text} phx-change="value-changed" />
A handle_event/3 implementation with the name "value-changed" will be called anytime the user changes the text of the TextField.
def handle_event("value-changed", new_value, socket) do
  {:noreply, assign(socket, text: new_value)}
end
The phx-debounce, phx-throttle, and phx-target attributes can be used to configure the event.
Refer to the documentation for an element to find out if it supports client-side changes.

  
    
  
  Server-side changes


Each element has an attribute that controls its value.
For example, TextField in the SwiftUI client uses the attribute text.
See the documentation for an element to find out what attribute it uses.
Whenever the attribute's value is changed, the client will update to display the new value.
No change event is sent when the server updates the value.

  
    
  
  Modifier change events


Some modifiers have values that can be changed by the client.
To receive change events on these modifiers, pass a LiveViewNativePlatform.Modifier.Types.Event to the change argument.
sheet(is_presented: @show, change: "presentation-changed")
Provide a map to change for more advanced configuration.
sheet(is_presented: @show, change: %{ event: "presentation-changed", debounce: 2000, target: @myself })
A change event will be sent with every argument of the modifier that can be change-tracked.
def handle_event("presentation-changed", %{ "is_presented" => show }, socket) do
  {:noreply, assign(socket, show: show)}
end
In the example above, the server can show/hide the sheet by setting the show assign.
When the user swipes down on the sheet to close it, the "presentation-changed" event will be called with the value false.
See the documentation for a modifier to find out which arguments can be change-tracked.



  

    
LiveViewNative 
    



      
A module providing supporting functions for LiveView Native.

      


      
        Summary


  
    Functions
  


    
      
        platform(platform_id)

      


        Returns an environment struct for a LiveView Native platform given its
platform_id or :error if not found.



    


    
      
        platform!(platform_id)

      


        Returns an environment struct for a LiveView Native platform given its
platform_id or raises if not found.



    


    
      
        platforms()

      


        Returns a list of environment structs for all LiveView Native platforms.



    





      


      
        Functions

        


  
    
      
      Link to this function
    
    platform(platform_id)



  


  

Returns an environment struct for a LiveView Native platform given its
platform_id or :error if not found.
Used to introspect platforms at compile-time or runtime.

  



  
    
      
      Link to this function
    
    platform!(platform_id)



  


  

Returns an environment struct for a LiveView Native platform given its
platform_id or raises if not found.
Same as platform/1 but raises RuntimeError instead of returning
:error if no platform exists for the given platform_id

  



  
    
      
      Link to this function
    
    platforms()



  


  

Returns a list of environment structs for all LiveView Native platforms.

  


        

      



  

    
LiveViewNative.Assigns 
    







  

    
LiveViewNative.Component 
    



      
Upgrades a Phoenix Component to a native Phoenix Component.
To use, inherit with use LiveViewNative.Component
like so:
Example:
defmodule MyApp.MyComponent do
  use MyAppWeb, :live_component
  use LiveViewNative.Component

  # ...
end

      





  

    
LiveViewNative.Layouts 
    




      
        Summary


  
    Functions
  


    
      
        compile_layout(arg1, template_path, opts)

      


    


    
      
        extract_layout(filename, opts)

      


    


    
      
        extract_layouts(opts)

      


    


    
      
        extract_layouts_recursive(nodes, opts)

      


    


    
      
        generate_class_trees(layouts, opts)

      


    


    
      
        layout_template(template_path, is_root_template?)

      


    


    
      
        layout_template_with_csrf_token(template, arg2)

      


    


    
      
        layout_template_with_live_reload(template, arg2)

      


    


    
      
        matches_template?(arg, filename)

      


    


    
      
        persist_class_trees(layouts, opts)

      


    





      


      
        Functions

        


  
    
      
      Link to this function
    
    compile_layout(arg1, template_path, opts)



  


  


  



  
    
      
      Link to this function
    
    extract_layout(filename, opts)



  


  


  



  
    
      
      Link to this function
    
    extract_layouts(opts)



  


  


  



  
    
      
      Link to this function
    
    extract_layouts_recursive(nodes, opts)



  


  


  



  
    
      
      Link to this function
    
    generate_class_trees(layouts, opts)



  


  


  



  
    
      
      Link to this function
    
    layout_template(template_path, is_root_template?)



  


  


  



  
    
      
      Link to this function
    
    layout_template_with_csrf_token(template, arg2)



  


  


  



  
    
      
      Link to this function
    
    layout_template_with_live_reload(template, arg2)



  


  


  



  
    
      
      Link to this function
    
    matches_template?(arg, filename)



  


  


  



  
    
      
      Link to this function
    
    persist_class_trees(layouts, opts)



  


  


  


        

      



  

    
LiveViewNative.LiveComponent 
    



      
Upgrades a Live Component to a native Live Component.
To use, inherit with use LiveViewNative.LiveComponent
like so:
defmodule MyApp.MyComponent do
  use MyAppWeb, :live_component
  use LiveViewNative.LiveComponent

  # ...
end

      





  

    
LiveViewNative.LiveSession 
    



      
Conducts platform detection on socket connections and applies
native assigns.

      


      
        Summary


  
    Functions
  


    
      
        on_mount(atom, params, session, socket)

      


    





      


      
        Functions

        


  
    
      
      Link to this function
    
    on_mount(atom, params, session, socket)



  


  


  


        

      



  

    
LiveViewNative.LiveView 
    



      
Upgrades a LiveView to a native LiveView.
To use, inherit with use LiveViewNative.LiveView
like so:
Example:
defmodule MyApp.MyLive do
  use MyAppWeb, :live_view
  use LiveViewNative.LiveView

  # ...
end

      





  

    
LiveViewNative.SessionPlug 
    




      
        Summary


  
    Functions
  


    
      
        call(conn, default)

      


    


    
      
        init(default)

      


    





      


      
        Functions

        


  
    
      
      Link to this function
    
    call(conn, default)



  


  


  



  
    
      
      Link to this function
    
    init(default)



  


  


  


        

      



  

    
LiveViewNative.Stylesheets 
    



      
Provides runtime support for LiveView Native's stylesheet system.

      


      
        Summary


  
    Functions
  


    
      
        get_class_tree(class_tree_module, tree_key \\ :default, opts \\ [])

      


        Returns a map containing all class names for templates within the given
LiveView, LiveComponent or Phoenix Components module. tree_key can be
passed for retrieving a specific class tree; this should only used for
modules which have multiple tree shaking passes applied at compile-time
(for example, layout modules), otherwise :default should be passed.



    


    
      
        get_class_tree_module(module)

      


        Returns the internal class tree module for the given LiveView, LiveComponent
or Phoenix Components module.



    


    
      
        reduce_stylesheets(map, stylesheet_modules)

      


        Compiles all stylesheets for the given class tree and returns them
as a single map.



    





      


      
        Functions

        


    

    

  
    
      
      Link to this function
    
    get_class_tree(class_tree_module, tree_key \\ :default, opts \\ [])



  


  

Returns a map containing all class names for templates within the given
LiveView, LiveComponent or Phoenix Components module. tree_key can be
passed for retrieving a specific class tree; this should only used for
modules which have multiple tree shaking passes applied at compile-time
(for example, layout modules), otherwise :default should be passed.

  



  
    
      
      Link to this function
    
    get_class_tree_module(module)



  


  

Returns the internal class tree module for the given LiveView, LiveComponent
or Phoenix Components module.

  



  
    
      
      Link to this function
    
    reduce_stylesheets(map, stylesheet_modules)



  


  

Compiles all stylesheets for the given class tree and returns them
as a single map.

  


        

      



  

    
LiveViewNative.TagEngine 
    



      
An implementation of Phoenix.LiveView.TagEngine that omits
HTML-centric template rules.

      


      
        Summary


  
    Functions
  


    
      
        classify_type(name)

      


        Callback implementation for Phoenix.LiveView.TagEngine.classify_type/1.



    


    
      
        void?(_)

      


        Callback implementation for Phoenix.LiveView.TagEngine.void?/1.



    





      


      
        Functions

        


  
    
      
      Link to this function
    
    classify_type(name)



  


  

Callback implementation for Phoenix.LiveView.TagEngine.classify_type/1.

  



  
    
      
      Link to this function
    
    void?(_)



  


  

Callback implementation for Phoenix.LiveView.TagEngine.void?/1.

  


        

      



  

    
LiveViewNative.Templates 
    



      
Provides functionality for preprocessing LiveView Native
templates.

      


      
        Summary


  
    Functions
  


    
      
        compile_class_tree(expr, platform_id, eex_opts)

      


    


    
      
        persist_class_tree_map(class_tree_map, caller)

      


    


    
      
        precompile(expr, platform_id, eex_opts)

      


    


    
      
        with_stylesheet_wrapper(expr, stylesheet_key \\ :default)

      


    





      


      
        Functions

        


  
    
      
      Link to this function
    
    compile_class_tree(expr, platform_id, eex_opts)



  


  


  



  
    
      
      Link to this function
    
    persist_class_tree_map(class_tree_map, caller)



  


  


  



  
    
      
      Link to this function
    
    precompile(expr, platform_id, eex_opts)



  


  


  



    

  
    
      
      Link to this function
    
    with_stylesheet_wrapper(expr, stylesheet_key \\ :default)



  


  


  


        

      



  

    
mix lvn.install 
    



      
Installer Mix task for LiveView Native: mix lvn.install

      


      
        Summary


  
    Functions
  


    
      
        infer_app_namespace(config_path)

      


    


    
      
        run(args)

      


        Callback implementation for Mix.Task.run/1.



    





      


      
        Functions

        


  
    
      
      Link to this function
    
    infer_app_namespace(config_path)



  


  


  



  
    
      
      Link to this function
    
    run(args)



  


  

Callback implementation for Mix.Task.run/1.

